Dinamica dei Sistemi Aziendali

Prova parziale del 31 ottobre 2025

Esercizio 1 (21 punti). Una società petrolifera possiede tre raffinerie, che possono produrre quattro diversi tipi di carburanti per auto: benzina, gasolio, benzina di alta qualità e gasolio di alta qualità. Le raffinerie possono lavorare al massimo rispettivamente 400, 600 e 800 tonnellate (t) di petrolio greggio a settimana. A causa dei diversi processi produttivi, a partire da una tonnellata di petrolio greggio si ottengono 0.9 t di benzina (o gasolio) e 0.8 t di benzina (o gasolio) di alta qualità. Si stima che il mercato richieda ogni settimana 500 t di benzina, 300 t di gasolio, 100 t di benzina di alta qualità e 50 t di gasolio di alta qualità. Il costo di lavorazione (in euro) di una tonnellata di petrolio greggio dipende sia dalla raffineria utilizzata che dal carburante prodotto ed è indicato nella seguente tabella:

Raffineria	Benzina	Gasolio	Benzina alta qualità	Gasolio alta qualità
1	6	5	8	7
2	7	6	9	8
3	8	7	10	9

Se una raffineria produce carburanti di alta qualità, è previsto un costo fisso aggiuntivo settimanale di 1000 euro. Per una raffineria, passare dalla produzione di un carburante non di alta qualità alla produzione di uno di alta qualità (o viceversa) richiede un certo tempo di *setup* che è preferibile evitare. Ogni raffineria può quindi produrre ogni settimana solo carburanti non di alta qualità oppure solo carburanti di alta qualità.

La società deve decidere ogni settimana quante tonnellate di petrolio greggio saranno utilizzate da ogni raffineria per la produzione di ogni carburante, rispettando i vincoli indicati sopra, con l'obiettivo di minimizzare il costo totale, dato dalla somma dei costi di lavorazione e dei costi fissi.

a) Scrivere un modello di Programmazione Lineare Intera per aiutare la società a risolvere il problema. Indicare le variabili decisionali, la funzione obiettivo ed i vincoli.

b) Scrivere nella sintassi di AMPL il modello di PLI formulato al punto a). Scrivere separatamente il file `.mod` ed il file `.dat`.

Esercizio 2 (10 punti). Considerare il problema di pianificare un progetto costituito da 6 attività che sono soggette a vincoli di precedenza. Il tempo di svolgimento (in giorni) ed i predecessori di ogni attività sono indicati nella seguente tabella:

Attività	A	B	C	D	E	F
Tempo di svolgimento	3	4	7	5	8	6
Predecessori	–	A	A,B	B,C	C,D	D,E

a) Rappresentare il problema di pianificare il progetto come un problema di cammino di costo massimo su un grafo opportuno.

b) Calcolare il minimo e massimo istante di inizio di ogni attività.

c) Calcolare la minima durata del progetto e trovare tutte le attività critiche.

Soluzioni

Esercizio 1.

a) Indichiamo le raffinerie con l'indice $i = 1, 2, 3$ ed i carburanti con l'indice j (benzina $j = 1$, gasolio $j = 2$, benzina di alta qualità $j = 3$, gasolio di alta qualità $j = 4$).

Definiamo le seguenti variabili:

x_{ij} = numero di tonnellate di petrolio utilizzate dalla raffineria i per produrre il carburante j,

$$y_i = \begin{cases} 1 & \text{se la raffineria } i \text{ produce carburanti di alta qualità,} \\ 0 & \text{altrimenti,} \end{cases}$$

$$z_i = \begin{cases} 1 & \text{se la raffineria } i \text{ produce carburanti non di alta qualità,} \\ 0 & \text{altrimenti,} \end{cases}$$

per ogni $i = 1, 2, 3$ e $j = 1, 2, 3, 4$.

Indicando con c_{ij} il costo di lavorazione di una tonnellata di petrolio greggio per produrre il carburante j nella raffineria i, il problema si può formulare mediante il seguente modello di PLI:

$$\begin{aligned}
&\min \sum_{i=1}^{3}\sum_{j=1}^{4} c_{ij}x_{ij} + 1000\sum_{i=1}^{3} y_i && && (1)\\
&x_{11} + x_{12} + x_{13} + x_{14} \leq 400 && && (2)\\
&x_{21} + x_{22} + x_{23} + x_{24} \leq 600 && && (3)\\
&x_{31} + x_{32} + x_{33} + x_{34} \leq 800 && && (4)\\
&0.9(x_{11} + x_{21} + x_{31}) \geq 500 && && (5)\\
&0.9(x_{12} + x_{22} + x_{32}) \geq 300 && && (6)\\
&0.8(x_{13} + x_{23} + x_{33}) \geq 100 && && (7)\\
&0.8(x_{14} + x_{24} + x_{34}) \geq 50 && && (8)\\
&x_{ij} \leq 800 z_i && \forall\, i = 1,2,3,\ j = 1,2 && (9)\\
&x_{ij} \leq 800 y_i && \forall\, i = 1,2,3,\ j = 3,4 && (10)\\
&y_i + z_i = 1 && \forall\, i = 1,2,3 && (11)\\
&x_{ij} \geq 0 && \forall\, i = 1,2,3,\ j = 1,\dots,4 &&\\
&y_i, z_i \in \{0,1\} && \forall\, i = 1,2,3 &&
\end{aligned}$$

Nella funzione obiettivo la prima somma rappresenta il costo di lavorazione, mentre la seconda i costi fissi dovuti alla produzione di carburanti di alta qualità.

I vincoli (2)–(4) garantiscono che la capacità produttiva di ogni raffineria sia rispettata.

I vincoli (5)–(8) garantiscono che la domanda settimanale di carburanti sia soddisfatta.

I vincoli (9)-(10) legano tra loro le variabili x, y e z.

I vincoli (11) garantiscono che ogni raffineria produce o solo carburanti di alta qualità o solo carburanti non di alta qualità.

b) Nella sintassi di AMPL il file `.mod` può essere scritto nel modo seguente:

```
#----------- parametri ---------------------

param n integer > 0;
```

```
# numero di carburanti

set CnoAQ within {1..n};
# insieme dei carburanti non di alta qualità

set CAQ within {1..n};
# insieme dei carburanti di alta qualità

param m integer > 0;
# numero di raffinerie

param CostoLav{i in 1..m, j in 1..n} >= 0;
# CostoLav[i,j] = costo di lavorazione di una t di petrolio greggio
# per produrre il carburante j nella raffineria i

param CostoFisso >= 0;
# costo fisso aggiuntivo settimanale
# se in uno stabilimento vengono prodotti carburanti di alta qualità

param Capacita{i in 1..m} >= 0;
# Capacita[i] = numero max di t di petrolio greggio che possono essere lavorate
# ogni settimana nella raffineria i

param Domanda{j in 1..n} >= 0;
# Domanda[j] = numero di t di carburante j richiesto ogni settimana dal mercato

param CoefnoAQ > 0;
# da una t di petrolio greggio si ottengono CoefnoAQ t di carburante non di alta qualità

param CoefAQ > 0;
# da una t di petrolio greggio si ottengono CoefAQ t di carburante di alta qualità

#------------ variabili ------------------

var x{i in 1..m, j in 1..n} >= 0;
# x[i,j] = numero di t di petrolio greggio utilizzate
# dalla raffineria i per produrre il carburante j

var y{i in 1..m} binary;
# y[i] = 1 se la raffineria i produce carburanti di alta qualità,
#        0 altrimenti

var z{i in 1..m} binary;
# z[i] = 1 se la raffineria i produce carburanti non di alta qualità,
#        0 altrimenti

#------------ funzione obiettivo ------------------

minimize CostoTotale: sum{i in 1..m, j in 1..n} CostoLav[i,j]*x[i,j]
+ sum{i in 1..m} CostoFisso*y[i];

#------------ vincoli ------------------

s.t. VCapProd{i in 1..m}: sum{j in 1..n} x[i,j] <= Capacita[i];
```

```
s.t. VDom1{j in CnoAQ}: sum{i in 1..m} CoefnoAQ*x[i,j] >= Domanda[j];

s.t. VDom2{j in CAQ}: sum{i in 1..m} CoefAQ*x[i,j] >= Domanda[j];

s.t. VProdnoAQ{i in 1..m, j in CnoAQ}: x[i,j] <= Capacita[i]*z[i];

s.t. VProdAQ{i in 1..m, j in CAQ}: x[i,j] <= Capacita[i]*y[i];

s.t. Valt{i in 1..m}: y[i] + z[i] = 1;
```

mentre il corrispondente file `.dat` può essere scritto come segue:

```
param n := 4;

set CnoAQ := 1, 2;

set CAQ := 3, 4;

param m := 3;

param CostoLav: 1 2 3 4 :=
1 6 5 8 7
2 7 6 9 8
3 8 7 10 9;

param CostoFisso := 1000;

param Capacita :=
1 400
2 600
3 800;

param Domanda :=
1 500
2 300
3 100
4 50;

param CoefnoAQ := 0.9;

param CoefAQ := 0.8;
```

Esercizio 2.

a) Il problema di trovare la durata minima del progetto equivale a trovare un cammino di costo massimo dal nodo o al nodo d sul seguente grafo:



b) Per trovare il minimo e massimo istante di inizio di ogni attività, applichiamo il metodo CPM. Un ordinamento topologico dei nodi è il seguente:



Il minimo e massimo istante di inizio di ogni attività sono riportati nella tabella seguente:

Attività	Minimo istante inizio	Massimo istante inizio
1	0	0
2	0	0
3	3	3
4	7	7
5	14	14
6	19	19
7	27	27
8	33	33

c) La minima durata del progetto è 33 giorni e tutte le 6 attività sono critiche.